
Towards A Certified Version of the Encyclopedia of Trian-
gle Centers

Julien Narboux and David Braun

Abstract. Triangle centers such as the center of gravity, the circumcenter, the orthocenter are well
studied by geometers. Recently, under the guidance of Clark Kimberling, an electronic encyclope-
dia of triangle centers (ETC) has been developed, it contains more than 7000 centers and many
properties of these points. In this paper, we describe how we created a certified version of ETC such
that some of the properties described come along with a computer checked proof of its validity.

Introduction

The characteristic points of triangles such as the center of gravity, the circumcenter, the orthocenter
have been well studied for centuries by geometers. Some properties of these points are well known.
For example the three points we cited are collinear, they belong to a line called the Euler line. More
recently, under the guidance of Clark Kimberling [Kim93, Kim94, Kim98], an electronic encyclopedia
of triangle center (ETC) has been developed1. This encyclopedia contains currently the definition of
more than 7000 triangle centers as well as the description of many properties of these points. Similarly,
an encyclopedia of more than 700 cubics is maintained by Bernard Gibert2 and an encyclopedia of
quadri-figures is maintained by Chris van Tienhoven3.

The ETC consists in web pages containing hundreds of thousands of properties. But most of the
results are presented without proof nor reference. A large majority of the results have been generated
by computer programs using symbolic and/or numeric manipulations. But the source code of the
computations are not given in the encyclopedia so the reader can not reproduce the results and we
can not verify if the result has been found by symbolic or numeric computations. How can one trust
theses results ? What are the potential sources of errors ?

1. There could be an error in the definition of the points in the Computer Algebra System (CAS)4.
2. There could be an error in the definition of the algorithms to check properties implemented in

the language of the CAS.
3. There could be a bug in the implementation of the CAS itself or in the theory behind the

algorithms used.

ICube, UMR 7357 CNRS, University of Strasbourg.
1http://faculty.evansville.edu/ck6/encyclopedia/ETC.html
2http://bernard.gibert.pagesperso-orange.fr/
3http://www.chrisvantienhoven.nl/index.php/mathematics/encyclopedia
4For example the definitions of points X5598 to X5602 contained an error in the version of ETC as of May 2015, the
point was defined as a4 − a2(b + c)2 − 4aS

√
r + 4R instead of a4 − a2(b + c)2 − 4aS

√
rR + 4R2

http://faculty.evansville.edu/ck6/encyclopedia/ETC.html
http://bernard.gibert.pagesperso-orange.fr/
http://www.chrisvantienhoven.nl/index.php/mathematics/encyclopedia

Number Name Trilinear Coordinates Barycentric Coordinates
X1 Incenter 1 a
X2 Centroid 1/a 1
X3 Circumcenter cos(A) a2(a2 − b2 − c2)
X4 Orthocenter sec(A) (a2 + b2 − c2)(a2 − b2 + c2)
X5 Nine-point center cos(B − C) a2b2 − b4 + a2c2 + 2b2c2 − c4

Table 1. Some ETC points

4. An error could have been introduced while copying the results from the computer software to
the web-page.

Our aim is to reduce these risks of errors. For this purpose we generate automatically the de-
scription of the points and we provide formal proofs. We created a certified version of ETC, such that
each property of a point come along with a computer checked proof of its validity with the necessary
non degeneracy assumptions.

The work closest to ours is the work of Deko Dekov who uses a computer program to generate
theorems in triangle geometry5 and present the results as a Computer Generated Encyclopedia of
Euclidean Geometry6. This work differs from ours because the proofs of the results are not given and
the facts are not checked using a proof assistant.

In the first section we provide an overview of the encyclopedia from the point of view of formal
theorem proving. In the second section we describe our formalization using the Coq proof assistant
of the required concepts. In the third section we explain how we obtained automatically thousands of
properties and how we proved them.

1. The Encyclopedia of Triangle Centers

The triangle centers defined in the encyclopedia of Clark Kimberling are various. The first centers
are well known points studied for centuries. The other points are defined by various means: most are
defined by a traditional geometric constructions, some are defined as minimizer of some geometric
quantity, there is even a point defined as a solution of an equation related to a physical property
(X5626 is the center of maximal electrostatic potential inside a triangle ABC having a homogeneous
surface charge distribution). Table 1 provides the list of the first 5 points in ETC. a, b and c denote
the side length of triangle ABC (a = |BC|, b = |AC| and c = |AB|). Figure 1 depicts the first 4000
points of ETC.

The points are defined by different means but are always manipulated using their homogeneous
coordinates relative to the triangle ABC. The coordinates used are either the barycentric coordinates
or the trilinear coordinates which are the relative distance from the point to the sides of the triangle
ABC. These coordinates are expressed using the length of the sides: a, b, c or equivalently the measure
of the three angles of triangle ABC. A given point P has trilinear coordinates any triple of the form
ka′ : kb′ : kc′ with k 6= 0 and a′, b′ and c′ denote the distances from P to the side lines of ABC
(Fig.2).

Kimberling defined a triangle center as a point X such that there is a function f of the side
lengths of triangle ABC such that: X = f(a, b, c) : f(b, c, a) : f(c, a, b) and f should be homogeneous,
there must be an integer n such that: ∀a b c t, t > 0 ⇒ f(ta, tb, tc) = tnf(a, b, c) and symmetric in
the sense that f(a, b, c) = f(a, c, b). The fact that f is a function of the side length ensure that the
triangle centers are preserved by isometries (as the side lengths are preserved by isometries). The fact
that f is homogeneous moreover ensures that triangle centers of similar triangles are preserved.

5http://www.ddekov.eu/j/contents.htm
6http://www.ddekov.eu/e1/

http://www.ddekov.eu/j/contents.htm
http://www.ddekov.eu/e1/

Figure 1. The first 4000 centers displayed (sketch generated using GeoGebra 5.0).

Figure 2. Trilinear Coordinates

To give to the reader an idea of what can be found in Clark Kimberling’s encyclopedia, we
reproduce here the entry of the orthocenter (Fig. 3). For most points in the encyclopedia the trilinear
and barycentric coordinates are given (often with several equivalent formulas). Then, some comments
are given about the point in natural language. Finally, we have a list of properties. The lines that the
point belongs to are listed. Then a list of relations between the points is given. For instance, X4 is the
midpoint of the segment X3X382. The amount of information available differs from one point to the
other.

Categories of points

We can distinguish triangle centers depending on how they can be expressed using trilinear coordinates,
this will be important to obtain proofs.

About 87% of the points can be defined by trilinear coordinates which are rational fractions in
a, b and c. The definition of quite a few points (about 12%) involve square roots, either square roots
of constants or also square roots of rational fractions in a, b and c. There are only a few occurrences of
nested square roots in the original list of coordinates collected by Peter Moses. One definition involve

a constant (
√

25− 10
√

5 in the definition of X1139) but the other occurrences can be simplified to

Figure 3. An example of one of the 7000 entries in Clark Kimberling’s encyclopedia:
point X4 orthocenter.

suppress the nested radicals. Some points (for instance X506 and X507) contain cube roots or fourth
roots (.1/3, .1/4) The definition of about 1% of the points contains trigonometric functions which can
not be eliminated : sin, cos, tan, sec, csc, sec, cot. For many points (such as the circumcenter or
orthocenter) the definition using trigonometric functions is short but they can be eliminated to ease
computations using the law of cosines (cos(A) = (b2 + c2 − a2)/(2bc)).

The definition of some points make use of the value of the angles (A, B, C) without using
any trigonometric function. For example X360, Hofstadter zero point, has trilinear coordinates: A/a :
B/b : C/c. For very few points (X368, X369, X370, X1144, X3232, X5373, X5626) we do not have the
explicit coordinates, thus we do not treat those points. In order to obtain smaller expressions for the
coordinates, some notations (known as Conway’s notations) are used, Tab. 2 lists those notations.

Symbol Description Definition

∆ Triangle area

√
(a+b−c)(a−b+c)(−a+b+c)(a+b+c)

4

SW a2+b2+c2

2

sa b+c−a
2

sb c+a−b
2

sc a+b−c
2

s Half-perimeter a+b+c
2

J
√
a6−a4b2−a2b4+b6−a4c2+3a2b2c2−b4c2−a2c4−b2c4+c6

abc

e
√

a4−a2b2+b4−a2c2−b2c2+c4

a2b2+a2c2+b2c2

R Radius of the circumcircle abc√
(a+b+c)(b+c−a)(c+a−b)(a+b−c)

r Radius of the incircle 1
2

√
(b+c−a)(c+a−b)(a+b−c)

a+b+c

Table 2. Conway’s notations.

2. Formalization of the main concepts using Coq

In this section, we describe the formalization in Coq of the concepts we need. The results given in
this section are stated and proved interactively using the Coq proof assistant (even if we use some
automatic tactics). In the next sections, we will describe the facts which have been found and proved
completely automatically.

For the Coq formalization we start with the type of the points, we use a record with three real
numbers to represent the homogeneous coordinates. We don’t specify in the type the exact kind of
coordinate we use, it can be either trilinear or barycentric coordinates. This type will also be used to
represent lines because of the duality between points and lines using homogeneous coordinates.

Record Point : Set :=

cTriple

{

X : R;

Y : R;

Z : R

}.

Note that a unique point can be represented using different homogeneous coordinates, hence we
define point equality as:

Definition eq_points P Q := X(Q)*Y(P) = X(P)*Y(Q) /\ X(Q)*Z(P) = X(P)*Z(Q).

We also define a type class representing the given base triangle defined by three real numbers:
the lengths of the sides a, b and c. We assume that the sides are strictly positive and verify the triangle
inequality.

Class triangle_theory := {

a:R;

b:R;

c:R;

apos: a > 0;

bpos: b > 0;

cpos: c > 0;

ineq_1 : a + b - c > 0;

ineq_2 : a - b + c > 0;

ineq_3 : -a + b + c > 0

}.

In this setting, the three reference points can be defined as7:

Definition pointA := cTriple 1 0 0.

Definition pointB := cTriple 0 1 0.

Definition pointC := cTriple 0 0 1.

As many theorems are incorrect for points on the side lines of triangle ABC, to simplify our
statements we introduce a predicate:

Definition is_not_on_sidelines P :=

match P with

cTriple X Y Z => X<>0 /\ Y<>0 /\ Z<>0

end.

To ease the definition of triangle centers we define a constructor which take as an argument an
homogeneous function h and produce the associated triangle center:

Context ‘{M:triangle_theory}. (* We assume we have a triangle theory. *)

Definition cPointh h := cTriple (h a b c) (h b c a) (h c a b).

Our development uses trilinear coordinates, but the definition of the coordinates of the points that
Peter Moses kindly gave us, uses barycentric coordinates. Fortunately it is easy to convert from one
to the other. For more convenience we define an alternative constructor to obtain trilinear coordinates
from a function for barycentric coordinates:

Definition cPointhb h := cTriple ((h a b c)*b*c) ((h b c a)*a*c) ((h c a b)*a*b).

Using these constructors we define 6000 points, we list here the first five points:

Definition X_1 := let h_x_1 a b c := a in cPointhb h_x_1.

Definition X_2 := let h_x_2 a b c := 1 in cPointhb h_x_2.

Definition X_3 := let h_x_3 a b c := a^2*(a^2-b^2-c^2) in cPointhb h_x_3.

Definition X_4 := let h_x_4 a b c := (a^2+b^2-c^2)*(a^2-b^2+c^2) in cPointhb h_x_4.

Definition X_5 := let h_x_5 a b c := a^2*b^2-b^4+a^2*c^2+2*b^2*c^2-c^4 in cPointhb h_x_5

2.1. Incidence geometry

A very important predicate in geometry is the collinearity predicate. Using homogeneous coordinates
three points are collinear if their determinant is zero, we can define the col predicate as:

Definition col A B C :=

X(A)*Y(B)*Z(C) + Y(A)*Z(B)*X(C) + Z(A)*X(B)*Y(C)

- Z(A)*Y(B)*X(C) - Y(A)*X(B)*Z(C) - X(A)*Z(B)*Y(C) = 0.

The line going through P and Q can be defined as a triple (the three coefficients of the equation
of the line):

7We name the points like that to keep the letters A B C for other purposes.

Definition line P Q :=

match P,Q with

cTriple x1 y1 z1, cTriple x2 y2 z2 =>

cTriple (y1*z2-y2*z1) (z1*x2-z2*x1) (x1*y2-x2*y1)

end.

Using the duality, the point of intersection of two lines can use the same definition, but we
provide an alternative definition to increase readability:

Definition inter L M := line L M.

2.2. Euclidean Geometry

Two lines of coordinates (x1, y1, z1) and (x2, y2, z2) are parallel if

x1 y1 z1

x2 y2 z2

a b c

= 0:

Definition parallel L M :=

match L,M with

cTriple x1 y1 z1, cTriple x2 y2 z2 =>

x1*y2*c+x2*b*z1+a*y1*z2 -a*y2*z1-x1*b*z2-x2*y1*c = 0

end.

We can then easily prove the triangle midpoint theorem to check our definitions:

Lemma triangle_midpoint : forall A B C,

parallel (line (midpoint A B) (midpoint A C)) (line B C).

Now, we can define a predicate for perpendicular lines. We adapt the formula which can be found
in the literature [Gal10] to get rid of trigonometric functions and divisions:

Definition perp L M :=

match L,M with

cTriple x1 y1 z1, cTriple x2 y2 z2 =>

2*a*b*c*(x1*x2 + y1*y2 + z1*z2)

-(y1*z2 + y2*z1)* a* (b^2+c^2-a^2)

-(z1*x2 + z2*x1)* b* (c^2+a^2-b^2)

-(x1*y2 + x2*y1)* c* (a^2+b^2-c^2) = 0

end.

Then, we can check that the circumcenter (X3 in Kimberling’s numbering) belongs to the per-
pendicular bisectors:

Lemma circumcenter_is_on_perp_bisect :

perp (line X_3 (midpoint pointA pointB)) (line pointA pointB).

We define the foot of the altitudes:

Definition T_A := inter (line pointB pointC) (line pointA X_4).

Definition T_B := inter (line pointA pointC) (line pointB X_4).

Definition T_C := inter (line pointA pointB) (line pointC X_4).

With these definitions we can check that the orthocenter is properly defined:

Lemma orthocenter_check :

perp (line X_4 T_A) (line pointB pointC) /\

Figure 4. Cevian triangle and pedal triangle.

perp (line X_4 T_B) (line pointA pointC) /\

perp (line X_4 T_C) (line pointA pointB).

2.3. About triangles

Cevian and anti-Cevian triangles. The concept of Cevian triangles will be useful in the rest of this
paper. In this paragraph we define the Cevian and anti-Cevian triangles of a point P . We adopt a
definition based on coordinates and we will prove afterward that the definition is correct. Recall the
Cevian triangle of a point P in a triangle ABC is the triangle formed by the intersections of the three
lines joining the vertexes of ABC to P and the sides of ABC (Fig. 4). A′B′C ′ is the anti-Cevian
triangle of P in ABC, if ABC is the Cevian-triangle of P in ABC.

Definition cevian_triangle P :=

match P with cTriple p q r =>

(cTriple 0 q r, cTriple p 0 r, cTriple p q 0)

end.

Definition anti_cevian_triangle P :=

match P with cTriple p q r =>

(cTriple (-p) q r, cTriple p (-q) r, cTriple p q (-r))

end.

But these definitions allow us to define only the Cevian triangle of P w.r.t. the base triangle. If
we want to consider the Cevian-triangle of P in any triangle ABC, we need a more general definition:

Definition cevian_triangle_gen P A B C :=

((inter (line A P) (line B C)),

(inter (line B P) (line A C)),

(inter (line C P) (line A B))).

We can convince ourselves that particular case for the base triangle ABC is consistent with the
general definition:

Lemma cevian_triangle_ok : forall P,

eq_triangles (cevian_triangle P) (cevian_triangle_gen P pointA pointB pointC).

We can prove also that we really defined the Cevian triangle:

Lemma cevian_triangle_gen_ok :

forall P A B C,

match cevian_triangle_gen P A B C with

(X,Y,Z) => col A P X /\ col B P Y /\ col C P Z /\

col X B C /\ col Y A C /\ col Z A B

end.

Finally, we can name some well known triangles:

Definition incentral_triangle := cevian_triangle X_1.

Definition medial_triangle := cevian_triangle X_2.

Definition orthic_triangle := cevian_triangle X_4.

Definition intouch_triangle := cevian_triangle X_7.

Definition extouch_triangle := cevian_triangle X_8.

Pedal triangle. The Pedal triangle APBPCP of a point P is the triangle formed by the projections of
P on the side lines (Fig.4), we can define its coordinates:

Definition pedal_triangle P :=

match P with cTriple p q r =>

(cTriple 0 (q+p*(-a^2-b^2+c^2)/(-2*a*b)) (r + p*(-a^2+b^2-c^2)/(-2*a*c)),

cTriple (p + q*(-a^2-b^2+c^2)/(-2*a*b)) 0 (r+q*(a^2-b^2-c^2)/(-2*b*c)),

cTriple (p + r*(-a^2+b^2-c^2)/(-2*a*c)) (q+r*(a^2-b^2-c^2)/(-2*b*c)) 0)

end.

As previously, we prove afterward that our definition make sense:

Lemma pedal_triangle_ok : forall P,

match pedal_triangle P with

(A1,B1,C1) => perp (line P A1) (line pointB pointC) /\

perp (line P B1) (line pointA pointC) /\

perp (line P C1) (line pointA pointB)

end.

Obviously, the Cevian triangle of the orthocenter is the same as the pedal triangle:

Lemma cevian_triangle_of_orthocenter_is_pedal :

eq_triangles (cevian_triangle X_4) (pedal_triangle X_4).

Moreover, the medial triangle is the pedal triangle of the circumcenter:

Lemma pedal_triangle_of_circumcenter_is_medial :

eq_triangles medial_triangle (pedal_triangle X_3).

The intouch triangle is the pedal triangle of the incenter:

Lemma pedal_triangle_of_incenter_is_intouch :

eq_triangles intouch_triangle (pedal_triangle X_1).

Perspective triangles. Two triangles ABC and A′B′C ′ are said to be perspective from O if their three
pairs of corresponding vertexes are joined by lines which meet in O (Fig. 5):

Definition is_perspector O T1 T2 :=

match T1,T2 with

(A,B,C),(A’,B’,C’) =>

Figure 5. Perspective triangles.

col O A A’ /\ col O B B’ /\ col O C C’

end.

For example, P is the perspector of any triangle ABC and its Cevian triangle in ABC:

Lemma cevian_triangle_perspector :

forall A B C P, is_perspector P (cevian_triangle_gen P A B C) (A,B,C).

2.4. The geometric transformations

We define the geometric transformations using homogeneous coordinates. Table 3 lists all the trans-
formations we can deal with. The definitions of these transformations are taken from Kimberling’s
glossary or Paul Yiu’s book [Yiu02]. We could have given the definitions using geometric construc-
tions, but we adopt algebraic definitions for two reasons. First, with algebraic definition, the proofs
are relatively easy to obtain. Second, the geometric definitions often require a proof to certify the
definition make sense. For example, the geometric definition of the centroid requires to prove that
the medians intersect. Still, we want to make sure we did not make any mistake while giving our
definitions. To check our definitions, we prove the geometric properties.

Ceva conjugate. As a first example, let’s consider the P -Ceva conjugate of Q. By definition it is the
perspector of the Cevian triangle of P and the anti-Cevian triangle of Q but we define it as:

Definition ceva_conjugate P U :=

match P,U with

cTriple p q r, cTriple u v w =>

cTriple (u*(-q*r*u+r*p*v+p*q*w)) (v*(q*r*u-r*p*v+p*q*w)) (w*(q*r*u+r*p*v-p*q*w))

end.

To prove that our definition is consistent with the geometric definition, we can prove that the
P -Ceva conjugate of Q is the perspector of the Cevian triangle of P and anti-Cevian triangle of Q
(Fig. 6):

Lemma ceva_conjugate_property :

forall P Q, is_perspector (ceva_conjugate P Q) (cevian_triangle P) (anti_cevian_triangle Q).

Name#Arity Definition

anticomplement#1 −ap+bq+cr
a

ap−bq+cr
b

ap+bq−cr
c

ceva conjugate#2 u(−qru + rpv + pqw) v(qru− rpv + pqw) w(qru + rpv − pqw)

complement#1 bq+cr
a

ap+cr
b

ap+bq
c

cross conjugate#2 u
−pvw+qwu+ruv

v
pvw−qwu+ruv

w
pvw+qwu−ruv

cross difference#2 qw − rv ru− pw pv − qu

cross sum#2 qw + rv ru + pw pv + qu

cross point#2 pu(rv + qw) qv(pw + ru) rw(qu + pv)

daleth conjugate#2 let h(p, q, r, u, v, w) = p( v
q −

w
r )2 + u( 2u

p −
v
q −

w
r ) in

h(p, q, r, u, v, w) h(q, r, p, v, w, u) h(r, p, q, w, u, v)

hirst inverse#2 qru2 − vwp2 rpv2 − wuq2 pqw2 − uvr2

iso conjugate#2 qrvw rpwu pquv

isogonal conjugate#1 1
p

1
q

1
r

isotomic conjugate#1 1
pa2

1
qb2

1
rc2

line conjugate#2 p(v2 + w2)− q(w2 + u2)− r(u2 + v2)−

u(qv + rw) v(rw + pu) w(pu + qv)

midpoint#2 let h = ap + bq + cr in

let k = au + bv + cw in

kp + hu kq + hv kr + hw

reflection#2 (ap− bq − cr)u+ (bq − ap− cr)v+ (cr − ap− bq)w+

2p(bv + cw) 2q(au + cw) 2r(au + bv)

waw conjugate#2 let h(p, q, r, x, y, z) = p(x2q2r2 + 2p2(ry − qz)2 − pqr2xy − pq2rxz) in

h(p, q, r, u, v, w) h(q, r, p, v, w, u) h(r, p, q, w, u, v)

zayin conjugate #2 let h(p, q, r, x, y, z) = p(y + z)2 − ry2 − qz2 + (p− r)xy + (p− q)xz in

h(p, q, r, u, v, w) h(q, r, p, v, w, u) h(r, p, q, w, u, v)

Table 3. The transformations formalized.

Isotomic conjugate. A second example is the isotomic conjugate. Given a point P , let A1B1C1 be the
Cevian triangle of P in ABC. Let A2 (resp. B2, C2) be reflection of A1 (resp. B1,C1) in the midpoint
of segment BC (resp. AC,AB). The lines AA2, BB2 and CC2 meet at a point P ′ called the isotomic
conjugate of P (Fig. 7). We define it using coordinates as:

Definition isotomic_conjugate P :=

match P with

cTriple p q r =>

cTriple (1/(p*a^2)) (1/(q*b^2)) (1/(r*c^2))

end.

Figure 6. The Ceva conjugate.

Figure 7. The isotomic conjugate.

Then to prove that our definition is correct, we express the fact that if we reflect the Cevian
triangle of P in the midpoints of the sides of ABC we obtain the Cevian triangle of its isotomic
conjugate. Note that we need to assume that P is not on any sideline of triangle ABC.

Lemma isotomic_conjugate_property :

forall P,

is_not_on_sidelines P ->

match cevian_triangle P with

(A1,B1,C1) =>

match cevian_triangle (isotomic_conjugate P) with

(A2,B2,C2) => eq_points (midpoint pointA pointB) (midpoint C1 C2)

end

Easily we can show also that this transformation is an involution:

Lemma isotomic_involution : forall P,

is_not_on_sidelines P ->

eq_points (isotomic_conjugate (isotomic_conjugate P)) P.

Complement and anti-complement. U is the complement of a point P if it belongs to line X2P (where
X2 is the centroid) and X2 trisects segment PU and X2 is closer to P than U . We define it as:

Definition complement P :=

match P with

cTriple p q r =>

cTriple ((b*q+c*r)/a) ((a*p+c*r)/b) ((a*p+b*q)/c)

end.

Definition anticomplement P :=

match P with

cTriple p q r =>

cTriple ((-a*p+b*q+c*r)/a) ((a*p-b*q+c*r)/b) ((a*p+b*q-c*r)/c)

end.

We can prove that they are inverse function of each-other:

Lemma anticomplement_complement_property :

forall P, eq_points (anticomplement (complement P)) P.

and that the centroid belongs to the line:

Lemma complement_property :

forall P, col P (complement P) X_2.

Reflection and midpoint. Following the formulas given in Tab. 3, we can define the midpoint of a
segment and the reflection of point wrt. another point, then we can check that:

Lemma reflection_midpoint : forall A B,

eq_points (midpoint A (reflection A B)) B.

As an example, we can also prove that the center of the nine point circle is the midpoint of the
segment X3X4 (X3 is the circumcenter and X4 the orthocenter):

Lemma nine_point_circle_bisects_OH:

eq_points X_5 (midpoint X_3 X_4).

3. Finding conjectures and automating their proofs

In the previous section, we listed our definitions and presented some results that we proved interactively
just to verify our definitions. In this section we describe the algorithms we used to find and prove
automatically some properties about triangle centers.

3.1. Finding properties

There are so many properties presented in Clark Kimberling’s encyclopedia that obtaining an exhaus-
tive list of these properties is difficult. Indeed, it would require to parse the information presented on
the web pages, but the computer generated pieces of information are inter-mixed with human gener-
ated comments in such a way that it would make this parsing phase non-trivial. Hence, we chose not
to check the properties displayed on the page, but to generate ourself these properties. Fortunately

most of the properties displayed in the encyclopedia are relatively easy to obtain by computer means.
In this section, we describe how we obtained these properties.

We address two main kind of problems:

1. Find all the collinearity properties in the list of points. Namely find all the triples (i, j, k) such
that Col(Xi, Xj , Xk).

2. Given a function f which takes n points as arguments and return a point, find all n + 1-tuples
(i, j1, . . . , jn) such that Xi = f(Xj1 , . . . , Xjn).

We want to find properties which are valid in any triangle, but to simplify the computations
we consider one example triangle. If the property does not hold in this triangle, it does not hold in
general, if it holds in this particular triangle then there are good chance that it holds in general. In
practice the coincidences are very rare (but Francisco Javier recently pointed that points X3875 and
X7292 have exactly the same coordinate in the triangle (a, b, c) = (6, 9, 13)).

3.1.1. Finding Collinearity Properties. Let n be the number of points in ETC. We want to find all
the triples (i, j, k) such that Col(Xi, Xj , Xk). A näıve algorithm would require O(n3) operations. We
implemented a better algorithm by using an associative array T to store for each line the set of points
belonging to this line. For all every couple of points (Xi, Xj) we compute the equation e of the line
XiXj and we add the points Xi and Xj to T [e]. This algorithm has a complexity O(n2) on average.

The main difficulty concerns the numerical computations. Our first prototype implemented using
MapleTMwas producing both false positive and false negative: some points appeared to be collinear
whereas there were not (even in the special case of the chosen triangle) and also some points which
are collinear were reported not to be collinear. This is due to the fact that MapleTM(or other similar
software) does not compute itself the precision needed to obtain a significant result. In a second proto-
type, we use what we call a symbolic-numeric approach. Instead of representing the coordinates of the
points using big-floats approximations, we used symbolic expressions involving constants, arithmetic
operations (+,-,*,/), square roots and trigonometric functions. Deciding equality of two expressions

involving radicals is possible but not trivial: for example
√

3 + 2
√

2 = 1 −
√

2. To decide such an
equality involving only constants we use the radnormal function of MapleTM. But we are not aware
of any algorithm dealing with the trigonometric functions. To prove the facts, we need an algorithm
to normalize symbolic expressions containing radicals, an algorithm has been proposed by Rybow-
icz which generalize an algorithm introduced by Zippel [Zip85, Ryb03] but we are not aware of any
implementation in a CAS.

3.1.2. Finding Properties About Transformations. To find the properties about transformations we
use a brute-force algorithm. Recall that, given a function f which takes n points as arguments and
return a point, we need to find all n + 1-tuples (i, j1, . . . , jn) such that Xi = f(Xj1 , . . . , Xjn). For
every tuple of points (Xj1 , . . . , Xjn) we compute X = f(Xj1 , . . . , Xjn) and we check if X belongs to
the (sorted) list of points. The complexity of this algorithm is O(knln(k)) where k is the number of
points in the encyclopedia.

3.2. Proving the properties

To prove the different properties in Coq, we make heavy use of the ring and field tactics of Coq
first introduced by Delahaye and Mayero and improved by Grégoire and Mahboubi [DM01, GM05].
The idea of these reflexive tactics is the following: instead of convincing Coq that two expressions
representing the same polynomial are equal by repeated applications of commutativity, associativity
and distributivity properties, an algorithm to compute a normal form is written in Coq’s language
and proved correct. For more information about how to develop reflexive tactics see Chapter 16 of
[BC04].

Proving properties involving only points whose coordinates are rational fractions is straightfor-
ward, we just need to unfold all the definitions to obtain a goal which can be solved by the field tactic.
We do not treat the points whose coordinates involve trigonometric functions. For the points with
square roots in their coordinates we use the following heuristics:

Name Number of facts proved Number of conjectures Percentage

ceva conjugate 7592 13091 58%
complement 906 932 97%
cross conjugate 2249 5483 41%
cross sum 4989 8552 58%
cross point 3381 8052 41%
daleth conjugate 8939 16216 55%
hirst inverse 9130 16383 56%
isogonal conjugate 1905 1954 97%
isotomic conjugate 604 616 98%
line conjugate 40136 57074 70%
waw conjugate 83 805 10%

total 79914 129158 62%
Table 4. Statistics

1. We try to simplify square roots by applying the following lemma repeatedly (where S n stands
for n + 1):

Lemma sqrt_pow : forall x n, 0 <= x ->

(sqrt x)^(S (S n)) = x*(sqrt x)^n.

2. We normalize all notations before unfolding them. For each notation, we have a lemma stating
a permutation property, such as:

Lemma Delta_perm_1 : DeltaMaj a b c = DeltaMaj b c a.

3. We normalize all square roots of rational fractions appearing in the expression. We have a tactic
normalize all sqrt which computes the normal form of every square root appearing in the
goal. Here is an example of its use:

Goal sqrt ((a+b)^2) + sqrt(a+b) = sqrt(b+a)+sqrt(a^2+b^2+2*a*b).

Proof.

normalize_all_sqrt.

ring.

Qed.

Non degeneracy conditions. The non degeneracy conditions needed for proving the properties
are that all the denominators appearing in the statements are different from zero and the the
argument of any square root is positive, they can be generated automatically.

3.3. Results

Figure 8 shows a snapshot our website displaying the pieces of information about the point X1. Each
fact is displayed in three colors:

Green. We have a proof verified by the Coq proof assistant of this fact (assuming the non degeneracy
conditions).

Orange. We have a MapleTMscript to check this fact.
Red. This result is only a conjecture obtained by numerical computations.

Table 4 shows the number of conjectures and the number of properties proved. It takes about 20
computer.days for Coq to compile these files.

Figure 8. A snapshot of the first entry of the Certified ETC.

Conclusion

We presented our first attempts toward a certified version of the encyclopedia of triangle centers. This
experiment shows that the approach based on trilinear coordinates is efficient and can be used within a
proof assistant. Compared to other automated deduction approaches such as Wu’s or Gröbner’s bases
methods [Kap86, CG90, Wan95], this approach consists in keeping always the explicit coordinates of
the points, in some sense the triangulation of the algebraic system is computed in advance. This saves
a lot of computations in this context where we want to prove many properties about the same given
points. This work also demonstrates that the implementation of the Coq proof assistant scales, and
can manage thousands of definitions.

This work can be extended in several directions.

First, it would be interesting to connect our proof about trilinear coordinates to our development
in the context of the geometry of Tarski [Nar07, BN12, BNSB14]. For example we could prove that the
definition of the collinearity predicate given in this paper is equivalent with the geometric definition
given in the context of Tarski’s axiom system.

Second, we could study how to find and prove automatically more properties. For example we
could enhance our algorithm to find properties of some well known triangles, such as ’the incenter is
the circumcenter of the intouch triangle’.

Third, we could improve the user interface. It would be interesting to have an interactive website
allowing the user to ask questions in a simple syntax, to find automatically if a theorem is already
known and to propose additions to the encyclopedia.

Fourth, we could generalize this work and try to certify the results presented in the encyclopedia
of conics and encyclopedia of quadri-figures.

Availability

The website is available here:
http://dpt-info.u-strasbg.fr/~narboux/CETC/about.html

Thanks

We thank Peter Moses for providing a file with barycentric coordinates of the points. We thank Pascal
Schreck for his comments about a draft of this paper.

